Problem:
Score : 500 points

Problem Statement

AtCoDeer is thinking of painting an infinite two-dimensional grid in a checked pattern of side K.
Here, a checked pattern of side K is a pattern where each square is painted black or white so that each connected component of each color is a K × K square.
Below is an example of a checked pattern of side 3:

AtCoDeer has N desires.
The i-th desire is represented by x_i, y_i and c_i.
If c_i is B, it means that he wants to paint the square (x_i,y_i) black; if c_i is W, he wants to paint the square (x_i,y_i) white.
At most how many desires can he satisfy at the same time?

Constraints

1 ≤ N ≤ 10^5

1 ≤ K ≤ 1000

0 ≤ x_i ≤ 10^9

0 ≤ y_i ≤ 10^9

If i ≠ j, then (x_i,y_i) ≠ (x_j,y_j).

c_i is B or W.

N, K, x_i and y_i are integers.

Input

Input is given from Standard Input in the following format:

N K
x_1 y_1 c_1
x_2 y_2 c_2
:
x_N y_N c_N

Output

Print the maximum number of desires that can be satisfied at the same time.

Sample Input 1

4 3
0 1 W
1 2 W
5 3 B
5 4 B

Sample Output 1

4

He can satisfy all his desires by painting as shown in the example above.

Sample Input 2

2 1000
0 0 B
0 1 W

Sample Output 2

2

Sample Input 3

6 2
1 2 B
2 1 W
2 2 B
1 0 B
0 6 W
4 5 W

Sample Output 3

4

Solution in Haskell (GHC 7.10.3) (judge verdict: Accepted)

{-# OPTIONS_GHC -O2 -funbox-strict-fields #-}
import           Control.Monad         (forM_)
import qualified Data.Array.ST         as MA
import           Data.Array.Unboxed    (UArray, (!))
import qualified Data.Array.Unboxed    as A
import qualified Data.ByteString.Char8 as BS
import           Data.Maybe            (fromJust)

main :: IO ()
main = do
  [n, k] <- readInts <$> BS.getLine
  xycs <- map readInt2Char . BS.lines <$> BS.getContents
  print $ solve n k xycs

solve :: Int -> Int -> [(Int, Int, Char)] -> Int
solve n k xycs = maximum [max c (n - c) | x <- [1 .. k], y <- [1 .. k], let c = count x y] where
  xys = map (convert k) xycs
  xys' = A.accumArray (+) 0 ((0, 0), (2 * k - 1, 2 * k - 1))
    [((x, y), 1) | (x, y) <- xys] :: UArray (Int, Int) Int
  bss = cumSum2 xys'
  count x y = count' 0 0 x y
    + count' (x + k) 0 (2 * k) y
    + count' x y (x + k) (y + k)
    + count' 0 (y + k) x (2 * k)
    + count' (x + k) (y + k) (2 * k) (2 * k)
  count' x y x' y' = bss ! (x', y') - bss ! (x, y') - bss ! (x', y) + bss ! (x, y)

cumSum2 :: UArray (Int, Int) Int -> UArray (Int, Int) Int
cumSum2 xss = MA.runSTUArray $ do
  let ((w0, h0), (w1, h1)) = A.bounds xss
  yss <- MA.newArray ((w0, h0), (w1 + 1, h1 + 1)) 0
  forM_ [w0 + 1 .. w1 + 1] $ \i -> do
    forM_ [h0 + 1 .. h1 + 1] $ \j -> do
      let x = xss ! (i - 1, j - 1)
      a <- MA.readArray yss (i - 1, j)
      b <- MA.readArray yss (i, j - 1)
      c <- MA.readArray yss (i - 1, j - 1)
      MA.writeArray yss (i, j) $ x + a + b - c
  return yss

convert :: Int -> (Int, Int, Char) -> (Int, Int)
convert k (x, y, 'B') = (x `mod` (2 * k), y `mod` (2 * k))
convert k (x, y, 'W') = convert k (x, y + k, 'B')
convert _ _           = undefined

readInt :: BS.ByteString -> Int
readInt = fst . fromJust . BS.readInt

readInts :: BS.ByteString -> [Int]
readInts = map readInt . BS.words

readInt2Char :: BS.ByteString -> (Int, Int, Char)
readInt2Char = (\[a, b, c] -> (readInt a, readInt b, BS.index c 0)) . BS.words